
Int. J. Appl. Math. Comput. Sci., 2002, Vol.12, No.4, 559–569

GENERATING QUASIGROUPS FOR CRYPTOGRAPHIC APPLICATIONS

A method of generating a practically unlimited number of quasigroups of a (theoretically) arbitrary order using the com-
puter algebra system Maple 7 is presented. This problem is crucial to cryptography and its solution permits to implement
practical quasigroup-based endomorphic cryptosystems. The order of a quasigroup usually equals the number of characters
of the alphabet used for recording both the plaintext and the ciphertext. From the practical viewpoint, the most important
quasigroups are of order 256, suitable for a fast software encryption of messages written down in the universal ASCII code.
That is exactly what this paper provides: fast and easy ways of generating quasigroups of order up to 256 and a little more.

1. Introduction

According to eminent specialists (Dénes and Keedwell,
1999) the earliest use of quasigroups in cryptography is
attributed to the German mathematician R. Shauffler, who
reduced in his Ph.D. dissertation of 1948 the problem of
breaking the Vigènere cipher to determining a particu-
lar latin square. In the above source, publications (Koś-
cielny, 1995; 1996), concerning explicitly the application
of quasigroups for construction stream and block ciphers,
are also quoted as rather rare serious contemporary efforts
to utilize the simplest non-associative algebraic systems in
cryptography. In this context the work (Ritter, 1998) is by
all means also worth noticing.

The paper presents a method of generating a prac-
tically unlimited number of quasigroups of an arbitrary
order (practically≤ 256) by means of the computer alge-
bra system Maple 7, for applications to cryptography. The
mathematical background knowledge, mandatory for un-
derstanding this article, is rather extensive. It is assumed
that the reader is familiar with combinatorial structures
and knows the basic properties of groups and Galois fields,
and that he or she is acquainted with the basics of number
theory. The only essential topics concerning latin squares
and quasigroups are given in Section 2, since these math-
ematical constructions are not generally known. Section 3
briefly comments the obtained results, while the routines,
implemented in Maple 7, for producing latin squares and
quasigroups, are presented in Appendix.

2. Latin Squares and Quasigroups

A latin square of ordern is an n× n array in whichn2

symbols, taken from a setA, are arranged so that each
symbol occurs only once in each row and exactly once
in each column. For any positive integern there exists
a latin square of ordern. A latin square of ordern for
which

A = {0, 1, . . . , n− 1} (1)

is said to be normalized or reduced if the elements of both
its first row and its first column are in a natural order. For
eachn ≥ 2 the total numberLS(n) of latin squares of
order n (Laywine and Mullen, 1998) is given by

LS(n) = n! (n− 1)! T (n), (2)

whereT (n) denotes the number of reduced latin squares
of order n.

The number of latin squaresLS(n) of order n in-
creases very quickly withn and is indeed great, even
for rather smalln. It should be noted that the number
of reduced latin squares is exactly known forn ≤ 10
(McKay and Rogoyski, 1995) (see Table 1). Forn =
11, 12, . . . , 15, T (n) is estimated in Table 2.

For n > 15 the bounds ofLS(n) can be calculated
(Jacobson and Matthews, 1996) using the formula

n∏
k=1

(k!)
n
k ≥ LS(n) ≥ (n!)2n

nn2 . (3)

In Table 3 some upper and lower bounds ofLS(n) for
several values ofn, most frequently used in practice and
obtained by means of (3), are given.

Table 1. Number of reduced latin squaresT (n) vs. n.

n T (n)

2 1

3 1

4 4

5 56

6 9 048

7 16 942 080

8 535 281 401 585

9 377 597 570 964 258 816

10 7 580 721 483 160 132 811 489 280

Table 2. Estimates ofT (n) for n = 11, 12, 13, 14, 15.

n T (n)

11 5.36 · 1033

12 1.62 · 1044

13 2.51 · 1056

14 2.33 · 1070

15 1.50 · 1086

Table 3. Estimates ofLS(n) for n = 2k, k = 4, 5, 6, 7, 8.

0.689 · 10138 ≥ LS(16) ≥ 0.101 · 10119,

0.985 · 10784 ≥ LS(32) ≥ 0.414 · 10726,

0.176 · 104169 ≥ LS(64) ≥ 0.133 · 104008,

0.164 · 1021091 ≥ LS(128) ≥ 0.337 · 1020666,

0.753 · 10102805 ≥ LS(256) ≥ 0.304 · 10101724.

As can be seen (Kościelny, 1995; 1997; Kóscielny
and Mullen, 1999), in the case of quasigroup-based stream
ciphers, a latin square is an essential element for the im-
plementation of both a key generator and the encipher-
ing/deciphering procedure. Since the number of latin
squares of orderq (usually 32 ≤ q ≤ 256), equal to the
number of elements of an alphabet used in practice to rep-
resent the plain-text and the cryptogram, is immense, the
key space is practically unlimited. Therefore, quasigroup-
based ciphers are much more efficient and more secure
than those based on regular algebraic systems such as
fields or groups, or on deterministic algorithms (elliptic
curve cryptosystems, conventional public-key cryptosys-
tems).

It is justified to recall here that an algebraic system

Q = 〈A, •〉 (4)

consisting of a finite set of elementsA in which a binary

operation• is defined, is called a quasigroup if for any
two elementsa, b ∈ A each of the equations

a • x = b, y • a = b (5)

has exactly one solution (Dénes and Keedwell, 1974). A
quasigroup is then a simple algebraic system with one op-
eration, which must be neither commutative nor associa-
tive, but, according to (5), the operation table in a quasi-
group ought to be a latin square. A quasigroup

L = 〈A, •〉 (6)

in which there exists an identity elemente ∈ A with the
property that

∀x ∈ A [ e • x = x • e = x ] (7)

is called a loop. If the operation• is associative and if

∀a ∈ A ∃ ã ∈ A [ a • ã = ã • a = e ], (8)

then the loopL becomes a group (the elementã is called
the opposite ofa or the inverse ofa). Although the setA
can have elements of any kind, it is assumed in this paper
that its elements are non-negative integers, as in (1).

To obtain a multiplication table of ann-element
quasigroup, one must have a suitably bordered latin square
of order n. The bordering means that the latin square
must be supplemented with the first row, which enumer-
ates columns, and with the first column, which enumer-
ates rows. If, e.g., the following latin square of order 8 is
given:

LS =



0 1 2 3 4 5 6 7
1 0 3 2 5 4 7 6
2 3 0 1 6 7 4 5
3 2 1 0 7 6 5 4
4 5 6 7 0 1 2 3
5 4 7 6 1 0 3 2
6 7 4 5 2 3 0 1
7 6 5 4 3 2 1 0


,

then the multiplication table of the corresponding quasi-
group 〈{0, 1, 2, 3, 4, 5, 6, 7}, •〉 will be

• 0 1 2 3 4 5 6 7
0 0 1 2 3 4 5 6 7
1 1 0 3 2 5 4 7 6
2 2 3 0 1 6 7 4 5
3 3 2 1 0 7 6 5 4
4 4 5 6 7 0 1 2 3
5 5 4 7 6 1 0 3 2
6 6 7 4 5 2 3 0 1
7 7 6 5 4 3 2 1 0

. (9)

In the case considered the operation table in the group
isomorphic to an additive group ofGF (8) has been ob-
tained: the system〈{0, 1, 2, 3, 4, 5, 6, 7}, •〉 is simul-
taneously a quasigroup, a loop and a group. Therefore
any finite group provides a latin square to a cryptographer,
who may easily transform it into many quasigroups, using
the notion of isotopy explained below.

Let
Qp = 〈G, ◦〉 (10)

and
Qi = 〈H, ?〉 (11)

be two quasigroups with|G| = |H|. An ordered triple

(πx, πy, πt) (12)

of one-to-one mappingsπx, πy, πt of the setG onto
the setH is called anisotopism of Qp upon Qi if

πx(x) ? πy(y) = πt(x ◦ y) (13)

for all x, y ∈ G. It should be noted that the mappingπt

permutes the elements in the table of operations in a quasi-
group Qp, while πx and πy operate on the elements of
the row and column borders of this table, respectively. It
is also said thatQi is an isotope of a primary quasi-
group Qp. One can prove that the set of all isotopisms of
a quasigroup of ordern forms a group of order(n!)3.

From (13) it follows that

X ? Y = πt

(
π−1

x (X) ◦ π−1
y (Y )

)
(14)

for all X, Y ∈ H. Equation (14) allows us to operate on
elements ofQi if the table of operations inQp is known.
It is evident that if πx = πy = πt, then the algebraic
systems (10) and (11) are isomorphic.

The isotopy is then a practically inexhaustible source
of quasigroups, since without any problems we can
construct many groups of an arbitrary finite ordern,
and thereby many latin squares, and, using (14), gen-
erate (n!)3 quasigroups. For example, assume that
Qp = 〈{0, 1, 2, 3, 4, 5, 6, 7}, •〉 with the operation ta-
ble given by (9). If

π−1
x =

(
01234567
46275130

)
, π−1

y =

(
01234567
47106532

)
,

πt =

(
01234567
17246053

)
,

then, using (14), we obtain the quasigroupQp =
〈{0, 1, 2, 3, 4, 5, 6, 7}, ◦〉 with the operation table as

in (15):

◦ 0 1 2 3 4 5 6 7
0 1 4 0 6 2 7 3 5
1 2 7 3 5 1 4 0 6
2 5 0 4 2 6 3 7 1
3 4 1 5 3 7 2 6 0
4 7 2 6 0 4 1 5 3
5 0 5 1 7 3 6 2 4
6 3 6 2 4 0 5 1 7
7 6 3 7 1 5 0 4 2

. (15)

The reader can easily verify (assuming that the first
operand from the left is the number of a row and the sec-
ond operand from the left is the number of a column) that
(0 ◦ 2) ◦ 3 = 6, 0 ◦ (2 ◦ 3) = 0, which means that the
operation◦ is not associative. ThenQi is a real non-
commutative quasigroup, since the operation table (15) is
not symmetrical with respect to the main diagonal.

The notion of thesimple product of quasigroupsis
also very important for application cryptographers since,
if two quasigroups are given, it permits to construct a
quasigroup of the order equal to the product of the orders
of these quasigroups.

Let

Q1 = 〈R, •〉, Q2 = 〈S ◦〉,

R = {0, 1, 2, . . . , n1 − 1},

S = {0, 1, 2, . . . , n2 − 1}

(16)

be two arbitrary quasigroups. Then the simple productQ
of these quasigroups is the algebraic system

Q = Q1 ×Q2 = 〈T, ?〉, (17)

where T = {0, 1, . . . , n1n2 − 1}. To define the opera-
tion ? in the setT , let us first assume that

Tcp = R × S = {t0, t1, . . . , tn1n2−1} (18)

is the Cartesian product of the setsR and S. Further, let

ti = (ri, si), tk = (rk, sk),

i, k ∈ T, ri, rk ∈ R, si, sk ∈ S.
(19)

Noting that the quasigroup with elements belonging to the
set Tcp

Qcp = 〈Tcp, ∗ 〉 (20)

is the simple product of quasigroups (16), we can define
the operation? as follows:

ti ∗ tk =
(
(ri • rk), (si ◦ sk)

)
. (21)

Table 4. Operation table for the quasigroupQcp = 〈Tcp, ∗ 〉 = Q1 × Q2.

∗ (0, 0) (0, 1) (0, 2) (1, 0) (1, 1) (1, 2) (2, 0) (2, 1) (2, 2) (3, 0) (3, 1) (3, 2)
(0, 0) (3, 1) (3, 2) (3, 0) (0, 1) (0, 2) (0, 0) (1, 1) (1, 2) (1, 0) (2, 1) (2, 2) (2, 0)
(0, 1) (3, 0) (3, 1) (3, 2) (0, 0) (0, 1) (0, 2) (1, 0) (1, 1) (1, 2) (2, 0) (2, 1) (2, 2)
(0, 2) (3, 2) (3, 0) (3, 1) (0, 2) (0, 0) (0, 1) (1, 2) (1, 0) (1, 1) (2, 2) (2, 0) (2, 1)
(1, 0) (2, 1) (2, 2) (2, 0) (1, 1) (1, 2) (1, 0) (0, 1) (0, 2) (0, 0) (3, 1) (3, 2) (3, 0)
(1, 1) (2, 0) (2, 1) (2, 2) (1, 0) (1, 1) (1, 2) (0, 0) (0, 1) (0, 2) (3, 0) (3, 1) (3, 2)
(1, 2) (2, 2) (2, 0) (2, 1) (1, 2) (1, 0) (1, 1) (0, 2) (0, 0) (0, 1) (3, 2) (3, 0) (3, 1)
(2, 0) (1, 1) (1, 2) (1, 0) (3, 1) (3, 2) (3, 0) (2, 1) (2, 2) (2, 0) (0, 1) (0, 2) (0, 0)
(2, 1) (1, 0) (1, 1) (1, 2) (3, 0) (3, 1) (3, 2) (2, 0) (2, 1) (2, 2) (0, 0) (0, 1) (0, 2)
(2, 2) (1, 2) (1, 0) (1, 1) (3, 2) (3, 0) (3, 1) (2, 2) (2, 0) (2, 1) (0, 2) (0, 0) (0, 1)
(3, 0) (0, 1) (0, 2) (0, 0) (2, 1) (2, 2) (2, 0) (3, 1) (3, 2) (3, 0) (1, 1) (1, 2) (1, 0)
(3, 1) (0, 0) (0, 1) (0, 2) (2, 0) (2, 1) (2, 2) (3, 0) (3, 1) (3, 2) (1, 0) (1, 1) (1, 2)
(3, 2) (0, 2) (0, 0) (0, 1) (2, 2) (2, 0) (2, 1) (3, 2) (3, 0) (3, 1) (1, 2) (1, 0) (1, 1)

But we want to represent the simple product of quasi-
groups (16) as a quasigroup (17). To this end, we must
convert the setTcp into the setT . There are many ways
of doing such a conversion. The mapping

λ : tx → X, tx ∈ Tcp, X, x ∈ T, (22)

defined by the function

λ(tx) = λ(rx, sx) = n2rx + sx, (23)

λ−1(X) = tx =
(
iquo(X, n2), X modn2

)
, (24)

gives one of the simplest solutions, since taking account
of (21)–(24), we have

X ? Y = λ
(
λ−1(X) ∗ λ−1(Y )

)
. (25)

Finally,

X ? Y = n2

(
iquo(X, n2) • iquo(Y, n2)

)
+ (X modn2) ◦ (Y modn2). (26)

In (24) and (26)X, Y ∈ T , iquo(m, n) computes the
integer quotient ofm divided by n, operators +, mod and
multiplication by n2 denote the usual arithmetic opera-
tions, while • and ◦ are operations in quasigroupsQ1

and Q2, respectively.
To explain in detail the calculation of the simple

product of quasigroups, assume that

Q1 = 〈R, •〉, Q2 = 〈S ◦〉,

R = {0, 1, 2, 3}, S = {0, 1, 2}
and

• 0 1 2 3
0 3 0 1 2
1 2 1 0 3
2 1 3 2 0
3 0 2 3 1

◦ 0 1 2
0 1 2 0
1 0 1 2
2 2 0 1

.

In this case

Tcp =
{
(0, 0), (0, 1), (0, 2), (1, 0), (1, 1), (1, 2),

(2, 0), (2, 1), (2, 2), (3, 0), (3, 1), (3, 2)
}

and the quasigroupQcp = 〈Tcp, ∗ 〉 = Q1 × Q2 has the
operation table given in Tab. 4.

Thus, after the application of the mapping (22), the oper-
ation table in the quasigroupQ = Q1 × Q2 = 〈T, ? 〉 is
given by Tab. 5.

Table 5. Operation table for the quasigroup
Q = 〈T, ? 〉 = Q1 × Q2.

? 0 1 2 3 4 5 6 7 8 9 10 11
0 10 11 9 1 2 0 4 5 3 7 8 6
1 9 10 11 0 1 2 3 4 5 6 7 8
2 11 9 10 2 0 1 5 3 4 8 6 7
3 7 8 6 4 5 3 1 2 0 10 11 9
4 6 7 8 3 4 5 0 1 2 9 10 11
5 8 6 7 5 3 4 2 0 1 11 9 10
6 4 5 3 10 11 9 7 8 6 1 2 0
7 3 4 5 9 10 11 6 7 8 0 1 2
8 5 3 4 11 9 10 8 6 7 2 0 1
9 1 2 0 7 8 6 10 11 9 4 5 3

10 0 1 2 6 7 8 9 10 11 3 4 5
11 2 0 1 8 6 7 11 9 10 5 3 4

where
T = {0, 1, . . . , 11}.

It should be noted that the simple product of two
quasigroups is not, in general, commutative, and that it
can be easily generalized to an arbitrary number of quasi-
groups.

To generate quasigroups of an arbitrary order, a pow-
erful software tool for doing the mathematics is needed.
For that purpose the computer algebra system Maple 7 can
be used, since this software offers, among other things,
powerful and effective integer arithmetic and operations
in the domain of univariate polynomials over the inte-
gers modulon, allowing computations in the finite field
GF (pm). There are also other useful Maple packages,
defining all number theory functions and routines, return-
ing not only random permutations, but also all the ele-
ments of the combinatorial class of required size. Fur-
thermore, each algorithm can be intelligibly and concisely
written down by means the Maple interpreter, easily com-
prehensible by any mathematician and programmer. But
as regards quasigroups, Maple is very poor: the only pack-
age MOLS returns a list of up topm−1 mutually orthog-
onal latin squares of orderpm, which are, in fact, interi-
ors of tables of addition inGF (pm) with permuted rows.
Therefore, in Appendix several useful algorithms for gen-
erating quasigroups, written in the Maple 7 interpreter, are
presented.

3. Conclusions

The problem concerning the construction of quasigroups,
important for the practice of quasigroup-based ciphers,
has been resolved. The Maple 7 routines presented in Ap-
pendix can be used directly for generating quasigroups,
while two routines are auxiliary: the first computes Zech’s
logarithms needed for constructing addition and subtrac-
tion tables inSGF (q), the other can write an arbitrary
latin square into a text file. All the routines can be used
as tools for designing both the encrypting and decrypting
procedures of quasigroup-based stream-ciphers, and the
key generators for such ciphers as well.

If, in some applications, the presented routines
seemed to be slow, one can implement them in other pro-

Appendix

Maple 7 Routines for Generating Latin Squares

All the names of the routines presented here, generating directly latin squares, begin with the lettersls , which means
latin square.

gramming languages (e.g. in C, C++, Fortran 95, etc.),
equipped with the appropriate libraries for doing advanced
mathematics. The execution time can then be shortened
even up to 100 and more times.

It should also be noted that the routinefsgfz can
be considerably optimized and, for applications, it should
be written in a high level programming language rather
than in the Maple interpreter. To do it, one must be famil-
iar with the properties of Zech’s logarithms inSGF (q)
(Kościelny, 1995).

1. LSs Isomorphic with the Interior of the Operation Table of a Cyclic Group, of
a Multiplicative Group of GF (pmmm), and of an Additive Group of GF (pmmm)

The first one-argument routinelscg (cg denotingcyclic group) computes the interior of the table of addition modulon
and returns a latin square in the form of ann× n matrix:

> lscg := proc(q::posint)
local i, j, ls;
ls := array(1 .. q, 1 .. q);
for i to q do

for j to q do ls[i, j] := (i + j - 2) mod q end do
end do;
evalm(ls)

end proc:

The procedurelscg can be called with an arbitrary actual argumentq.

As is known, the order of a multiplicative group ofGF (pm) equalspm − 1. Therefore, using a routine which gen-
erates a latin square of orderq, isomorphic with the interior of the multiplicative group ofGF (pm), we must remember
that the conditionq + 1 = pm must be satisfied.

The one-argument routinelsmgff (mgff stands formultiplicative group of a finite field) uses the GF package. This
routine generates also a latin square of the orderq = 256, which is important for practice since, in this case,q + 1 is a
prime.

> lsmgff := proc(q::posint)
local i, j, ls, m, p, w, g;
w := ifactor(q + 1);
p := op(convert(w, list)[1]);
if nops(w) = 1 then m := 1 else m := op(op(w)[2]) end if;
if p^m <> q + 1 then error "q+1 must be a power of prime"
end if;
g := GF(p, m);
ls := array(1 .. q, 1 .. q);
for i to q do for j to q do ls[i, j] :=

g:-output(g:-‘*‘(g:-input(i), g:-input(j))) - 1
end do

end do;
evalm(ls)

end proc:

Similarly to the previous procedure, the two-argument routinelsagff (agff stands foradditive group of a finite
field) makes use of the GF package. Thus, the order of the returned latin square, isomorphic to the operation table in the
additive group ofGF (pm), is equal to the power of the primepm.

> lsagff := proc(p::prime, m::posint)
local ls, g, i, j, n;
g := GF(p, m);
n := p^m;
ls := array(1 .. n, 1 .. n);
for i from 0 to n - 1 do for j from 0 to n - 1 do

ls[i + 1, j + 1] :=
g:-output(g:-‘+‘(g:-input(i), g:-input(j)))

end do
end do;
evalm(ls)

end proc:

A2. Two LSs Isomorphic with the Interior of Addition and Subtraction Tables
of a Spurious Galois Field

Each SGF (q) can provide two latin squares being the interiors of addition and subtraction tables (it is clear, however,
that the table of addition inGF (q) of characteristic 2 does not differ from the subtraction table in this field).

To determine addition and subtraction tables inSGF (q), the appropriate Zech logarithm is needed. It can be
computed using the two-argument routinefsgfz (f for file, sgf for spurious Galois field,z for Zech’s logarithm). The
routine implements an algorithm similar to that in (Kościelny, 1995, p. 179) and returns all values of Zech’s logarithms
for SGF (q). The result of calculation is written in the text file namedfn . Each row of the file containsZ(x), x =
1, 2, . . . , q − 2 if q is even, orZ(x), x = 0, 1, 2, . . . , (q − 3)/2, (q + 1)/2, . . . , q − 2 if q is odd.

> fsgfz := proc(q::posint, fn::string)
local i, k, b, f, nd, nz, p, z, za, zs, q2, n, nc, q1, c, ip, z2f;

q1 := q - 1;
q2 := q - 2;
f := fopen(fn, WRITE);
nd := convert(nops(convert(q1, base, 10)), string);
if q mod 2 = 0 then n := 1/2*q2 else n := 1/2*q1 end if;
z := array(1 .. q2);
with(combstruct);
c := iterstructs(Combination(q2), size = n);
nz := 0;
for i to count(Combination(q2), size = n) do

p := convert(nextstruct(c), list);
ip := iterstructs(Permutation(p));
while not finished(ip) do

za := nextstruct(ip);
b := false;
if q mod 2 = 0 then

for k to n do b := b or za[k] = k end do;
if not b then

for k to n do
z[k] := za[k];
z[q1 - k] := (z[k] - k + q1) mod q1

end do;
zs := convert(z, set);
if nops(zs) = q2 and not member(0, zs)
then

nz := nz + 1;
for k to q2 do fprintf(f, cat("%", nd, "d "), z[k])
end do;
fprintf(f, "\n")

end if
end if

else
for k to n - 1 do b := b or za[k] = k - 1
end do;
if not b then

for k to n do z[k] := za[k] end do;
for k to n - 1 do z[q2 - k + 1] :=

(za[k + 1] - k + q1) mod q1
end do;
zs := convert(z, set);
if nops(zs) = q2 and not member(0, zs)
then

nz := nz + 1;
for k to q2 do fprintf(f, cat("%", nd, "d "), z[k])
end do;
fprintf(f, "\n")

end if
end if

end if
end do

end do;
fprintf(f, "%s%d%s %d", "Number of sgf(", q, ") equals to", nz);
fclose(f)

end proc:

Calling the routine as in the following statements:

> fsgfz(9,"d:\\latsqr\\z09"); fsgfz(10,"d:\\latsqr\\z10");

will create two text files, namedz09 andz10 , situated on the discd: in the directorylatsqr (this directory must exist on
the discd:).

After computing the set of all Zech’s logarithms for a desiredq, we can generate two latin squares of orderq for
any element of this set, using the procedurelssgf . The second formal parameter of this routine,z , is a list containing
the values of an arbitrary row of the file, created by the routinefsgfz .

> lssgf := proc(q::posint, z::list)
local q1, q2, zech, i, k, m, ni, S, lsa, lsb;
S := proc(x, y::nonnegint)

local d, mn;
if x = 0 or y = 0 then return x + y
elif x <= y then mn := x
else mn := y
end if;
d := abs(x - y);
if d = ni then return 0
else return 1 + ((mn - 1 + zech[d]) mod q1)
end if

end proc;
q1 := q - 1;
q2 := q1 - 1;
lsa := array(1 .. q, 1 .. q);
lsb := array(1 .. q, 1 .. q);
zech := array(0 .. q2);
zech[0] := 1;
if q mod 2 = 0 then

ni := 0; for i to q2 do zech[i] := z[i] end do
else

ni := 1/2*q1;
for i to ni do zech[i - 1] := z[i] end do;
for i from ni + 1 to q2 do zech[i] := z[i] end do

end if;
zech[ni] := 0;
for i to q do

for k to q do lsa[i, k] := S(i - 1, k - 1) end do
end do;
for i to q do for k to q do for m to q do

if lsa[i, k] = m - 1 then lsb[m, k] := i - 1

end if
end do

end do
end do;
evalm(lsa), evalm(lsb)

end proc:

The algorithm according to which the procedurefsgfz is implemented is not the fastest one. Therefore, it practi-
cally operates on the value of the actual parameterq < 20. But it has been observed (Kościelny, 1996) that the discrete
function Z(2x− 1) = q/2− x + 1, Z(2x) = x, x = 1, 2, . . . , (q − 2)/2 satisfies the conditions for Zech’s logarithm
of SGF (q) for an arbitrary evenq ≥ 6.

The above is taken into account by the routinelspsgf (psgf stands forparticular sgf), having one argument and
operating exactly in the same manner as the procedurelssgf .

> lspsgf := proc(q::posint)
local S, i, k, m, a, b, q1;

if q mod 2 = 1 or q < 6 then
error "q must be an even integer > 5"

end if;
S := proc(x, y)

local d, mn, z;
if x = 0 or y = 0 then return x + y end if;
if x <= y then mn := x else mn := y end if;
d := abs(x - y);
if d mod 2 = 0 then z := 1/2*d
else z := 1/2*q + 1/2*d - 1/2
end if;
mn := mn - 1;
if d = 0 then return 0
else return 1 + ((mn + z) mod q1)
end if

end proc;
a := array(1 .. q, 1 .. q);
b := array(1 .. q, 1 .. q);
q1 := q - 1;
for i to q do

for k to q do a[i, k] := S(i - 1, k - 1) end do
end do;
for i to q do for k to q do for m to q do

if a[i, k] = m - 1 then b[m, k] := i - 1
end if

end do
end do

end do;
evalm(a), evalm(b)

end proc:

A3. Routine for Computing the Simple Product of an Arbitrary Number of Quasigroups

The routinelsspr (spr corresponds tosimple product) with an arbitrary number of arguments can be applied to gener-
ating a simple product of an arbitrary number of quasigroups, equal to the number of arguments passed in the procedure
call. It contains a local proceduresp2 , which is based exactly on the formulae (16)–(26). The proceduresp2 com-
putes the simple product of two arbitrary quasigroups and its use by the main routine depends on the number of actual
parameters in the call to the procedurelsspr .



Cz. Kościelny568

> lsspr := proc()
local sp2, i, ltsq;
sp2 := proc(a::matrix, b::matrix)

local i, k, cp, n1, n2, n12, ls;
n1 := linalg[rowdim](a);
n2 := linalg[rowdim](b);
n12 := n1*n2;
cp := array(1 .. n12);
ls := array(1 .. n12, 1 .. n12);
for i from 0 to n12 - 1 do

cp[i + 1] := [iquo(i, n2), i mod n2]
end do;
for i to n12 do for k to n12 do ls[i, k] :=

n2*a[cp[i][1] + 1, cp[k][1] + 1]
+ b[cp[i][2] + 1, cp[k][2] + 1]

end do
end do;
evalm(ls)

end proc;
ltsq := args[1];
if nargs = 1 then return args[1] end if;
for i from 2 to nargs do ltsq := sp2(ltsq, args[i])
end do

end proc:

For example, the call

> lsspr(lsmgff(4),lscg(2), lsagff(2,1));

will compute the product of three quasigroups of orders 4, 2 and 2.

A4. Routine for Determining the Isotope of an Arbitrary Quasigroup

The one-argument procedurelsis (is stands forisotope) transforms any latin square into the interior of an operation
table in a quasigroup, according to (14). The required three permutations are generated at random by means of the
combinat package.

> lsis := proc(ls::matrix)
local i, k, n, pix_, piy_, pit, q;
n := linalg[rowdim](ls);
q := array(1 .. n, 1 .. n);
pix_ := combstruct[draw](Permutation(n));
piy_ := combstruct[draw](Permutation(n));
pit := combstruct[draw](Permutation(n));
for i to n do for k to n do

q[i, k] := pit[ls[pix_[i], piy_[k]] + 1] - 1
end do

end do;
evalm(q)

end proc;

Since the permutationspix_ , piy_ andpit are generated at random, any call, repeated several times, may give
different quasigroups. In order to obtain the same result in any call, a seed for the random number generator must be set.
For example, by repeating the instructions

> _seed := 142857: lsis(lsspr(lsmgff(4), lscg(4)));

the reader will obtain every time the same result.

A5. Routine for Writing LS to a File

The implementation of quasigroup-based cryptographic systems is generally done in languages other than the Maple
interpreter. In this case a quasigroup, determined by Maple, ought to be written to a file and then imported into other
encrypting/decrypting software. To achieve this, we must modify the procedurelsqg . Of course, there are many ways of
doing it. For example, the procedureflsqg computes the same quasigroup as the routinelsqg , but, instead of returning
a computed quasigroup, it writes it to a text file namedlsq and created in the current directory.

fls := proc(ls::matrix, fn::string)
local f, i, k, n, nc;

n := linalg[rowdim](ls);
nc := convert(nops(convert(n, base, 10)), string);
f := fopen(fn, WRITE);
for i to n do

for k to n do
fprintf(f, cat("%", nc, "d "), ls[i, k])

end do;
fprintf(f, "\n")

end do;
fclose(f)

end proc:

> _seed := 13: fls(lsis(lspsgf(16)[2]),"d:\\latsqr\\q16");

Finally, from among plenty of statements using the routinefls , cooperating with the others procedures of this
section in order to create a file containing the interior of the operation table in a quasigroup of order 256, several examples
are given:

> fls(lsis(lsspr(lspsgf(16)[1], lspsgf(16)[2])), "d:\\latsqr\\q256a");
> fls(lsspr(lsis(lsmgff(16)), lsis(lsagff(2,4))),"d:\\latsqr\\q256b");
> fls(lsis(lsspr(lsis(lspsgf(8)[2]), lsis(lsagff(2,5)))),"d:\\latsqr\\q256c");
> fls(lsis(lsspr(lsagff(2,5), lsagff(2,3))),"d:\\latsqr\\q256d");
> fls(lsis(lsspr(lspsgf(16)[2], lscg(4), lsis(lsmgff(4)))),"d:\\latsqr\\q256e");
> fls(lsis(lsspr(lspsgf(64)[1], lsis(lsagff(2,2)))),"d:\\latsqr\\q256f");
> fls(lsis(lsspr(lsis(lsagff(2,3)), lspsgf(32)[2])),"d:\\latsqr\\q256g");
> fls(lsis(lsspr(lspsgf(8)[1], lspsgf(8)[2],lscg(4))),"d:\\latsqr\\q256h");
> fls(lsis(lsagff(2,8)),"d:\\latsqr\\q256i");
> fls(lsis(lspsgf(256)[2]),"d:\\latsqr\\q256j");
> fls(lsis(lsmgff(256)),"d:\\latsqr\\q256k");
> fls(lsis(lsspr(lspsgf(16)[1],lspsgf(16)[1])),"d:\\latsqr\\q256l");
> fls(lsis(lsspr(lsmgff(16),lsmgff(16))),"d:\\latsqr\\q256m");
> fls(lsis(lscg(256)),"d:\\latsqr\\q256n");
> fls(lsis(lsspr(lsagff(2,7),lsmgff(2))),"d:\\latsqr\\q256o");
> fls(lsis(lsspr(lspsgf(32)[1],lscg(8))),"d:\\latsqr\\q256p");

As can be seen, all the procedures presented here provide poweful means for generating quasigroups.

